

Chapter 1

Woo Jia Hao

September 29, 2019

Contents

1	The Elements of Programming	2
1.1	Expressions	2
1.2	Naming and the Environment	4
1.3	Evaluating Combinations	5
1.4	Compound Procedures	7
1.5	The Substitution Model for Procedure Application	8
1.5.1	Applicative order vs normal order	9
1.6	Conditional Expressions and Predicates	10
1.6.1	Logical composition operators	12
1.7	Example: Square Roots by Newton's Method	13
1.7.1	Newton's method of successive approximation	13
1.8	Procedures as Black-Box Abstractions	14
1.8.1	Local names	14
1.8.2	Internal definitions and block structure	15
1.9	Procedures and the Processes They Generate	16
1.9.1	Linear Recursion and Iteration	16
1.10	Chapter 1 Exercises	20
1.10.1	Exercise 1.1	20
1.10.2	Exercise 1.2	21
1.10.3	Exercise 1.3	21
1.10.4	Exercise 1.4	21
1.10.5	Exercise 1.5	22
1.10.6	Exercise 1.6	22
1.10.7	Exercise 1.7	23
1.10.8	Exercise 1.8	24
1.10.9	Exercise 1.9	24
1.10.10	Exercise 1.10	25

1 The Elements of Programming

Programming language serves multiple purposes.

- Means for instructing a computer to perform tasks
- Serves as a framework to organise our ideas about **processes**

We need to look at how the language combines single ideas to form more complex ones - which is done via the following:

- **Primitive expressions** - simplest entities the language is concerned with
- **Means of combination** - compound elements are built from simpler ones
- **Means of abstraction** - compound elements can be named and manipulated as units

Two elements of programming:

- **Data** - "stuff" to manipulate
- **Procedures** - descriptions of rules for manipulating data
 - Also described as methods
 - Used for combining and abstracting procedures and data

1.1 Expressions

In Lisp, entering an **expression** into an interpreter will cause the interpreter to *evaluate the expression*.

Combining expressions representing numbers with expressions representing primitive procedures form compound expressions

- Represents the application of the procedure to those numbers

(+ 21 26 12 7 35)

The above is an example of an expression representing the application of procedures to those numbers.

- Formed by delimiting a list of expressions within parentheses in order

- **Operator** - leftmost element in the list
- **Operands** - other elements in the list
- **Value of combination** - apply procedure specified by the operator to the arguments
- **Arguments** - value of the operands

Prefix notation - placing operator to the left of the operands.

- Advantages:
- Accommodates procedures with arbitrary number of arguments

```
(+ 21 35 12 7)
>>> 75
(* 25 4 12)
>>> 1200
```

- No ambiguity since operator is always the leftmost element and the entire combination is delimited by the parentheses
- Allows combinations to be nested

```
(+ (* 3 5) (- 10 6))
>>> 19
```

- * No limit to depth of nesting and to the overall complexity of the expression the Lisp interpreter can evaluate

Pretty printing - each long combination is written so that the operands are aligned vertically which display the structure of the expression.

```
(+ (* 3 (+ (* 2 4) (+ 3 5))) (+ (- 10 7) 6))

(+ (* 3
    (+ (* 2 4)
        (+ 3 5)))
    (+ (- 10 7)
        6))
```

Order of interpretation - regardless of complexity of expression.

- Read expression from terminal
- Evaluate expression
- Print results
- Aka *read-eval-print loop*

1.2 Naming and the Environment

Using a name to refer to computational objects

- Name identifies a *variable* whose *value* is the object

```
(define size 2)
```

That has associated the name *size* to the value 2. We can use this name to refer to the value 2 from now on.

```
size
>>> 2
(* 5 size)
>>> 10
```

define is the simplest means of abstraction.

- Allows the use of simple names to refer to the results of compound operations

Due to the complex structure of computational objects, it becomes difficult to remember their details and this is where using names help.

- Through the use of names, we build larger programs from smaller procedures.

Environment - memory that stores the *name-value* pairs.

- The one we use now is known as the **global environment**.

1.3 Evaluating Combinations

The interpreter takes the following steps to evaluate a combination:

- Evaluate the subexpressions of the combination
- Apply the procedure that is the value of the operator to the arguments


Evaluation is *recursive* in nature, meaning that in order for the operation to occur, it must invoke itself.

- In this scenario, for evaluation to occur, it must first evaluate all expressions

```
(* (+ 2 (* 4 6))  
  (+ 3 5 7))
```

In the example above, the combination can be represented by a tree.

- Each combination is represented by a node with branches corresponding to the operator and the operands of the combination stemming from it
- **Terminal nodes** represent either operators or numbers
- Values of operands precolate upward
 - Moving from terminal nodes and combining at higher and higher levels
 - **Tree accumulation** - process of accumulating



Due to the recursive nature of evaluation, we end up evaluating expressions, not combinations.

- Values of numerals are the numbers that they name.
- Values of built-in operators are the machine instruction sequences that carry out the corresponding operations.
- Value of other names are the objects associated with those names in the environment.

For the above rules of evaluation, the second rule is a special case of the third rule - the symbols $+$ / $*$ are stored in the global environment and are associated with the sequence of machine instructions as their *values*.

The evaluation rule does not handle definitions - instead of applying `define` to two arguments such as `(define x 3)`, `define` associates `x` to the value of 3.

- Definitions are **not** combinations
- Exception to the general evaluation rule

Special forms - exceptions to the general evaluation rule.

- Each special form has its own evaluation rule

1.4 Compound Procedures

Procedure definitions - compound operations with names and referred to as a unit

To illustrate the idea of procedure definition, we think of the procedure as an instruction:

To square something, multiply it by itself

Then, we express that in our language as such:

```
(define (square x) (* x x))
```

We have created a *compound procedure* with the name *square*.

- The procedure represents the operation of multiplying something by itself
- Thing to multiplied has the local name *x*

The general form of procedure definition is:

```
(define (<name> <formal parameters>) <body>)
```

- *<name>* - symbol to be associated with the procedure definition in the environment
- *<formal parameters>* - names used within the body of the procedure to refer to the corresponding arguments of the procedure
- *<body>* - expression that yield the value of the procedure application when the formal parameters are replaced by the actual arguments to which the procedure is applied
- *<name>* and *<formal parameters>* are grouped within parantheses
- As they would be in an actual call to the procedure being defined.

With *square* defined, we can now use it:

```
(square 21)
>>> 441
(square (+ 2 5))
>>> 49
```

We can even use it as a building block in defining other procedures.

```
(define (sum-of-squares x y)
  (+ (square x) (square y)))

(sum-of-squares 3 4)
>>> 25
```

1.5 The Substitution Model for Procedure Application

The interpreter applies the same process as primitive procedures to procedure application.

The body of the procedure is evaluated with each formal parameter is replaced by the corresponding argument.

```
(f 5)
```

And say that **f** has the following definition, it gives new meaning to the above procedure call.

```
(define (f x) (sum-of-squares (+ x 1) (* x 2)))

(f 5)
(sum-of-squares (+ 5 1) (* 5 2))
```

As such, the problem is now the evaluations of a combination with two operands and an operator, **sum-of-squares**.

With the new expanded form, we evaluate the parameters to 6 and 10 respectively.

Then, after replacing **f** with its body definition of **sum-of-squares**, we will continue to substitute each procedure with its body - in this case now, we will substitute **sum-of-squares** with its body comprising of **square**.

```
(sum-of-squares 6 10)
(+ (square 6) (square 10))
```

Then, we apply the body of **square** to obtain our final step.


```
(+ (square 6) (square 10))  
(+ (* 6 6) (* 10 10))
```

And now that we are left with only primitive operations, we will finally reduce it.

```
(+ 36 100)  
>>> 136
```

This process applied is known as the *substitution model* for procedure application.

- Way of thinking of procedure application, not an overview of how interpreters work
- More than 1 evaluation model

1.5.1 Applicative order vs normal order

Evaluating all operators and operands and then applying the procedure to the arguments is not the only method of evaluation.

An alternative is to only evaluate operands until their values are needed.

- Substitute operand expressions for parameters until it obtained an expression involving only primitive operators and then perform evaluation

```
(f 5)  
(sum-of-squares (+ 5 1) (* 5 2))  
(+ (square (+ 5 1) (* 5 2)))  
(+ (* (+ 5 1) (+ 5 1)) (* (* 5 2) (* 5 2)))  
(+ (* 6 6) (* 10 10))  
(+ 36 100)  
>>> 136
```

The thing to note with this evaluation model is that some procedures might be evaluated twice, like `(+ 5 1)` and `(* 5 2)`.

Normal-order evaluation - "fully expand and then reduce"

- Contrast to **applicative-order evaluation** - "evaluate the arguments and then apply"

Lisp uses applicative-order evaluation.

- Due to additional efficiency obtained from avoiding repeated evaluations of the same expressions
- Normal-order evaluation becomes much more complicated to deal with after leaving the realm of procedures that can be modelled by substitution

1.6 Conditional Expressions and Predicates

Case analysis - construct where we make tests and perform different operations depending on the result of said test.

$$|x| = \begin{cases} x & \text{if } x > 0 \\ 0 & \text{if } x = 0 \\ -x & \text{if } x < 0 \end{cases}$$

For instance, the above declares the function of *absolute*. In order to replicate this in Lisp, we use a special form known as `cond`.

```
(define (abs x)
  (cond ((> x 0) x)
        ((= x 0) 0)
        ((< x 0) (- x))))
```

`cond` general form:

```
(cond (<p1> <e1>)
      (<p2> <e2>)
      ...
      (<pn> <en>))
```

- **Clauses** - paranthesized pairs of expressions `<p> <e>`

- **Predicate** - $\langle p \rangle$ - expression whose value is interpreted as **true** or **false**
- **Consequent expression** - $\langle e \rangle$ - value to be given if the matching predicate is **true**
- Evaluated in order of clauses, if **p1** is false, then it moves on to **p2** and so forth
- If none of the predicates are true, the value of **cond** is undefined

Alternative for writing absolute-value procedure:

```
(define (abs x)
  (cond ((< x 0) (- x))
        (else x)))
```

Expressed in English as

+begin_{quote} If **x** is less than 0 return -**x**; otherwise return **x** #+end_{quote}
else - used in place of a predicate in the final clause of a **cond**

- **cond** returns its value if all other clauses have been bypassed (all other predicates are false)

Another alternative way of writing absolute-value procedure:

```
(define (abs x)
  (if (< x 0)
      (- x)
      x))
```

if - restricted type of conditional used when there're precisely two cases in the case analysis.

- General form:

```
(if <predicate> <consequent> <alternative>)
```

- Evaluation - starts with $\langle predicate \rangle$, if **true**, return $\langle consequent \rangle$, else, return $\langle alternative \rangle$

1.6.1 Logical composition operators

Alongside `<`, `=`, `>`, there are other logical composition operators.

1. `and`

- Evaluates left-to-right order
- If any `<e>` evaluates to **false**, entire expression is **false**
- If all `<e>` evaluate to **true**, only then will expression be **true**
- Special form, not procedure

```
(and <e1> ... <en>)
```

```
(and (> x 5) (< x 10))
```

The above expression represents a condition that a number `x` must be in the range `5 < x < 10`.

2. `or`

- Evaluates left-to-right order
- If any `<e>` evaluates to **true**, the whole expression is **true**
- If all `<e>` evaluates to **false**, the whole expression is **false**
- Special form, not procedure

```
(or <e1> ... <en>)
```

```
(define (>= x y)
  (or (> x y) (= x y)))
```

3. `not`

- If `<e>` evaluates to **false**, expression is **true** and vice versa

```
(not <e>)
```

1.7 Example: Square Roots by Newton's Method

Conceptually, procedures are akin to mathematical functions. However, what sets procedures apart from mathematical functions is the fact that they have to be effective.

Mathematically, we can represent the square-root function as

$$\sqrt{x} = \text{the } y \text{ such that } y \geq 0 \text{ and } y^2 = x \quad (1)$$

However, the definition, while accurate mathematically, does not define a procedure computationally.

```
(define (sqrt x)
  (the y (and (>= y 0)
              (= (square y) x))))
```

Herein lies the contrast between (mathematical) functions and procedures is a reflection of the distinction between describing properties of things and describing how to do things.

- Also referred to as the distinction between *declarative knowledge* and *imperative knowledge*.
- In mathematics, we are concerned with declarative descriptions (what is)
- In computer science, we are concerned with imperative descriptions (how to)

Leading to the use and definition of **Newton's method of successive approximations**.

1.7.1 Newton's method of successive approximation

We start with a guess y for the value of the square root of a number x .

To obtain a better guess (closer to the actual square root), we use the following manipulations:

$$\text{Average } y \text{ with } \frac{x}{y} \quad (2)$$

This average then becomes the new guess or y and we continue till we hit a "good enough" criteria which would be the case where the guess is as close to the square root as possible, often within some minute fractional difference.

We represent this definition in lisp as follows:

```

(define (sqrt-iter guess x)
  (if (good-enough? guess x)
      guess
      (sqrt-iter (improve guess x)
                  x)))

(define (improve guess x)
  (average guess (/ x guess)))

(define (average x y)
  (/ (+ x y) 2))

(define (good-enough? guess x)
  (< (abs (- (square guess) x)) 0.001))

(define (sqrt x)
  (sqrt-iter 1.0 x))

```

1.8 Procedures as Black-Box Abstractions

`sqrt-iter` is recursive - procedure is defined in terms of itself.

Any large program can be dissected into parts.

- Each procedure accomplishes an identifiable task that can be used as a module in defining other procedures.
- We regard each sub-procedure as a "black box"
 - We are not concerned with *how* it works, we only care that it computes the result
 - Aka **procedural abstraction**
 - The procedure definition should be able to suppress detail
 - User does not need to have written the procedure but can still use the code like a blackbox.

1.8.1 Local names

Meaning of a procedure should be independent of the parameter names used by its author.

- The parameter name must be local to the body of the procedure

- If the parameters are not local to the bodies of the procedure, the parameter might be confusing to the developer/user of the procedure
- **Formal parameter** - special role in the procedure definition - it doesn't matter what name the formal parameter has.
 - This is known as a **bound variable**
 - Procedure definition *binds* its formal definition
 - Definition remains unchanged if a bound variable is consistently renamed throughout the definition.
 - If a variable is not bound, it is *free*
 - **Scope** - set of expressions for which a binding defines a name

1.8.2 Internal definitions and block structure

Some procedures might involve functions with the same name but have different implementations.

- To remedy this issue, we create a procedure to have internal definitions that are local to that procedure

```
(define (sqrt x)
  (define (good-enough? guess x)
    (< (abs (- (square guess) x)) 0.001))
  (define (improve guess x) (average guess (/ x guess)))
  (define (sqrt-iter guess x)
    (if (good-enough? guess x)
        guess
        (sqrt-iter (improve guess x) x)))
  (sqrt-iter 1.0 x))
```

- **Block structure** - solution for the name-packaging problem proposed
 - To further add to the solution, since **x** is bound in the definition of **sqrt**, the nested procedures are all in the scope of **x**
 - Don't need to pass **x** to each procedure - **x** is a free variable in the internal definitions
 - Aka *lexical scoping*

1.9 Procedures and the Processes They Generate

The knowledge we possess now is akin to understanding the rules of chess but having no experience playing the game.

- To build upon this knowledge, we have to be able to /visualize the process generated by various types of procedures.

Procedure - pattern for *local evaluation* of a computational process.

- Specifies how each stage of the process is built upon the previous stage
- Make statements about the *global* behaviour of a process whose local evolution has been specified by a procedure

1.9.1 Linear Recursion and Iteration

Factorial is defined as the following function:

$$n! = n \cdot (n - 1) \cdot (n - 2) \cdots 3 \cdot 2 \cdot 1 \quad (3)$$

A common pattern for calculating factorial is realising that $n!$ is equal to n times $(n - 1)!$ for any positive integer n .

$$n! = n \cdot [(n - 1) \cdot (n - 2) \cdots 3 \cdot 2 \cdot 1] = n \cdot (n - 1)! \quad (4)$$

We can translate this program to lisp:

```
(define (factorial n)
  (if (= n 1)
      1
      (* n (factorial (- n 1)))))
(factorial 6)
```

An alternate approach to computing factorials would be to describe a rule that computing $n!$ would state that we would first multiply 1 by 2, then take the result of that multiplication and multiply it to 3 and so forth till we reach n

- A more formal definition of the rule would be that we maintain a *running product*, together with a *counter* that counts from 1 to n .




Figure 1: Linear recursive process

- As the calculations progress, the following changes are made to the *running product* and *counter*

$$product \leftarrow counter \times product \quad (5)$$

$$counter \leftarrow counter + 1 \quad (6)$$

- We also stipulate that $n!$ is the value of the product when the counter exceeds n

```

(define (factorial n)
  (fact-iter 1 1 n))
(define (fact-iter product counter max-count)
  (if (> counter max-count)
      product
      (fact-iter (* counter product)
                  (+ counter 1)
                  max-count)))

```

Much like the other process defined, we will produce a specific tree for the way the process is executed:

```
(factorial 6)  ▷
(fact-iter 1 1 6)
(fact-iter 1 2 6)
(fact-iter 2 3 6)
(fact-iter 6 4 6)
(fact-iter 24 5 6)
(fact-iter 120 6 6)
(fact-iter 720 7 6)
720  ◁
```




Figure 2: Linear recursive process

While the two processes perform the same operation, they carry different **shapes** and evolve quite differently over time.

1. Linear recursive process

- For the first process, the substitution model reveals a shape of expansion followed by contraction
 - **Expansion** - occurs as the process builds up a chain of *deferred operations*
 - **Contractions** - occurs as the operations as the operations are performed
 - **Recursive process** - process characterized by a chain of deferred operations
 - Interpreter keeps track of the operations to be performed later on
 - * Length of the chain of deferred multiplications == the amount of information needed to keep track of it, growing linearly with n
 - * Aka **proportional to n**
 - This process is known as a *linear recursive process*
- "Hidden information" is maintained by the interpreter and not contained in program variables.
 - These indicate where the process is
 - The longer the chain of deferment, the more information must be maintained

2. Linear iterative process

- For the second process, the process does not grow and shrink
- At each step, we keep track of *product*, *counter* and *max-count*
- Aka *iterative process* - any process whose *state* can be summarized by a fixed number of *state variables* along with a *fixed rule* that describes how the state variables should be updated as the process moves from state to state
 - Might include an **optional** end test that specifies conditions in which the process should terminate
 - Aka *linear iterative process*

- Program variables provide a complete description of the state of the process at any point
 - If the process is stopped mid-way, we would have to resume computation by providing the values of the program variables before the program was terminated.
3. Process vs procedure Recursive *process* is different from recursive *procedure*
- **Recursive procedure** - syntactic fact that the procedure definition refers to the procedure itself
 - Aka, it's a procedure that is recursive
 - **Recursive process** - how the process evolves, not the syntax
 - For instance, with the second implementation of the **factorial** procedure, it is described as being a *recursive procedure* with a *linear process* due to the shape it takes on
 - The distinction is made due to the implementations of other languages
 - These languages are designed to take any interpretation of any recursive procedure to consume an amount of memory that grows with the number of procedure calls - regardless of whether or not the process described is *iterative*
 - These languages can only describe iterative processes using special-purpose "**looping-constructs**"
 - This is not how Racket/Scheme is implemented
 - **Tail-recursion** - process of executing an iterative process that is described by a recursive procedure in constant space
 - * Everything is expressed using ordinary procedure call mechanisms

1.10 Chapter 1 Exercises

1.10.1 Exercise 1.1

1. 10
2. 12
3. 8

- 4. 3
- 5. 6
- 6. 3
- 7. 4
- 8. 19
- 9. false
- 10. 4
- 11. 16
- 12. 6
- 13. 16

1.10.2 Exercise 1.2

Refer to code

1.10.3 Exercise 1.3

Refer to code

1.10.4 Exercise 1.4

```
(define (a-plus-abs-b a b)
  ((if (> b 0) + -) a b))
```

The procedure is used to add **a** and **b** together. The purpose of the **if** condition is to accomodate for negative values of **b** - effectively working with the absolute value of **b**.

For instance, if **b** = -8 and **a** = 9, the predicate of the **if** condition will evaluate to **true** and so the resulting operator will be -. Therefore, the evaluated expression will be (- 9 (- 8)). When expanded to normal form (9 - (-8)) = (17).

1.10.5 Exercise 1.5

```
(define (p) (p))

(define (test x y)
  (if (= x 0)
      0
      y))

(test 0 (p))
```

With applicative-order evaluation, the expression is first evaluated then each argument is applied - meaning that the expressions are evaluated as they appear, rather than waiting till the very end. Therefore, the expression will attempt to evaluate `p`, which in this case is a function call. Since `p` is recursive, the interpreter will be stuck in an infinite loop.

However, with normal-order evaluation, the expression is expanded first and is only evaluated when needed - meaning that the interpreter will not attempt to evaluate `(p)` before fully expanding the procedure definition of `test`. This leads to the `if` condition being evaluated first and having the expression return 0 instead of being stuck in an infinite loop.

```
; Applicative-order evaluation
(test 0 (p))
(test 0 (p))
...
(test 0 (p))

; Normal-order evaluation
(test 0 (p))
(if (= 0 0)
    0
    (p))
>>> 0
```

1.10.6 Exercise 1.6

```
(define (new-if predicate then-clause else-clause)
  (cond (predicate then-clause)
        (else else-clause)))
```

```
(define (sqrt-iter guess x)
  (new-if (good-enough? guess x)
    guess
    (sqrt-iter (improve guess x)
      x)))
```

To understand how this new function will compute the square roots, we need to first see how the function will be evaluated. For this, we apply the **applicative-order evaluation**, the same one that lisp uses.

When we run the code in our terminal, it doesn't return anything and is instead stuck processing it. So let's investigate why. Unlike the built in if statement, **new-if** is a procedure defined by the developer. This means that when evaluating the expression, we first evaluate the arguments of **new-if** before determining what **new-if** does, and this causes it to hang because we're never actually comparing the arguments of **new-if** since the procedure will continue to recurse.

1.10.7 Exercise 1.7

For small numbers, our limit is too large to allow for an accurate reading. If the guesses reach a certain limit that exceeds the built in 0.001 limit, we will get false positives that are not accurate enough enough.

For large numbers, our limit is far too small for the system to appropriately measure the square root within a decent period of time since it will continue to refine the square root till it hits the 0.001 limit.

The solution would be to modify **good-enough?** to look at the difference between iterations.

```
; Old version
(define (good-enough? guess x)
  (< (abs (- (square guess) x)) 0.001))

; New version
(define (good-enough? guess x)
  (< (abs (- (improve guess x) guess))
    (* guess 0.001)))
```

In the old version, we compare the original number to the square of the guess. However, this is too strict of a requirement for the guesses to be accurate. The new version rectifies this issue by factoring in two key components.

1. The size of the leeway or limit
2. How much of a fit the guess was

This way, we are more flexible with the way we determine the limit for what qualifies as a `good-enough?` guess.

1.10.8 Exercise 1.8

Refer to code

1.10.9 Exercise 1.9

```
(define (inc x)
  (+ x 1))

(define (dec x)
  (- x 1))

(define (+ a b)
  (if (= a 0) b (inc (+ (dec a) b))))

(define (+ a b)
  (if (= a 0) b (+ (dec a) (inc b))))
```

For the first implementation, the growth of `+` will look like (note that we omit the calculation of `dec` and take that the numbers decrease automatically):

```
(+ 4 5)
(inc (+ 3 5))
(inc (inc (+ 2 5)))
(inc (inc (inc (+ 1 5))))
(inc (inc (inc (inc (+ 0 5)))))
(inc (inc (inc (inc 5))))
(inc (inc (inc 6)))
(inc (inc 7))
(inc 8)
>>> 9
```

For this, we see that the first `+` is a recursive process described by a recursive procedure.

The growth of the second implementation of `+` looks like:


```

(+ 4 5)
(+ 3 6)
(+ 2 7)
(+ 1 8)
(+ 0 9)
(9)
>>> 9

```

For this, we see that the second + is an iterative process described by a recursive procedure.

1.10.10 Exercise 1.10

```

(define (A x y)
  (cond ((= y 0) 0)
        ((= x 0) (* 2 y))
        ((= y 1) 2)
        (else (A (- x 1) (A x (- y 1))))))

(A 1 10)

(A 1 10)
(A 0 (A 1 9))
(A 0 (A 0 (A 1 8)))
(A 0 (A 0 (A 0 (A 1 7))))
(A 0 (A 0 (A 0 (A 0 (A 1 6)))))
(A 0 (A 0 (A 0 (A 0 (A 0 (A 1 5))))))
(A 0 (A 0 (A 0 (A 0 (A 0 (A 0 (A 1 4)))))))
(A 0 (A 0 (A 0 (A 0 (A 0 (A 0 (A 0 (A 1 3)))))))
(A 0 (A 0 (A 0 (A 0 (A 0 (A 0 (A 0 (A 0 (A 1 2)))))))
(A 0 (A 0 (A 0 (A 0 (A 0 (A 0 (A 0 (A 0 (A 0 (A 1 1)))))))
(A 0 (A 0 (A 0 (A 0 (A 0 (A 0 (A 0 (A 0 (A 0 (A 0 2)))))))
(A 0 (A 0 (A 0 (A 0 (A 0 (A 0 (A 0 (A 0 (A 0 4)))))))
(A 0 (A 0 (A 0 (A 0 (A 0 (A 0 (A 0 8))))))
(A 0 (A 0 (A 0 (A 0 (A 0 (A 0 16))))))
(A 0 (A 0 (A 0 (A 0 (A 0 32))))
(A 0 (A 0 (A 0 (A 0 64))))
(A 0 (A 0 (A 0 128)))
(A 0 (A 0 256))
(A 0 512)
(1024)

```

```

(A 2 4)

(A 2 4)
(A 1 (A 2 3))
(A 1 (A 1 (A 2 2)))
(A 1 (A 1 (A 1 (A 2 1))))
(A 1 (A 1 (A 1 2)))
(A 1 (A 1 (A 0 (A 1 1))))
(A 1 (A 1 (A 0 2)))
(A 1 (A 1 4))
(A 1 (A 0 (A 1 3)))
(A 1 (A 0 (A 0 (A 1 2))))
(A 1 (A 0 (A 0 (A 0 (A 1 1)))))
(A 1 (A 0 (A 0 (A 0 2))))
(A 1 (A 0 (A 0 4)))
(A 1 (A 0 8))
(A 1 16)
...
(65536)

```